MWSUG 2019 - Paper IN-19

Time-To-Event Analysis in the Presence of Competing risks

Xiaoting Wu, Department of Cardiac Surgery, Michigan Medicine, Ann Arbor, MI;

Donald S. Likosky, Department of Cardiac Surgery, Michigan Medicine, Ann Arbor, MI.

ABSTRACT

Competing risks are common phenomena in time-to-event analysis. A competing risk may take place before the event of interest thus exclude the possibility of event occurrence. For example, in the study of artificial heart valve duration, death is a competing risk as it modifies a patient's chance to receive potential reoperation due to valve deterioration. Ignoring competing risks, for example, the use of standard Kaplan-Meier estimators, will result in biased estimates for the event of interest. Cumulative incidence function that estimates the probability of event of interest over time, and cause-specific hazard function that models the effect of covariates on the event of interest, are two main approaches to perform time-to-event analysis in the presence of competing risk. This paper demonstrates the rational, implementation and interpretation of these methods, with SAS applications using SAS macro % CIF, LIFETEST and PHREG procedure.

INTRODUCTION

Competing risks may occur before the event of interest that preventing the occurrence of potential events [1, 2]. For example, death causes such as stroke, cancer, organ failure is competing events, such that only one of them can occur. In the study of elderly population, death is often a competing risk for other study events. For example, when analyzing the risk of transferring to a nursing home in the older age group, death is a competing risk event and need to be considered. In our data example, when studying the risk of reoperation after arch replacement surgery, death is a competing risk, because occurrence of death will preclude patients' chance of receiving a reoperation.

Instead of Kaplan Meier curve, cumulative incidence curve can be used to describe the event incidence over time when competing risk is present [1, 2].

$$CIF_k(t)=Pr(T \le t, D=k)$$

The function $CIF_k(t)$ can be interpreted as the probability of kth event before time t and before the occurrence of other type of events.

There are several methods to analysis competing risk through hazard function.

In the absence of competing risks, the hazard function is

$$\lambda(t) = \lim_{\Delta t \to 0} \frac{Prob\ (t\ \le T < t + \Delta t\ |T \ge t)}{\Delta t}$$

In the presence of competing risk, the following two hazards are of interest:

1. Fine and Gray subdistribution [3]

$$\lambda_k^{sd}(t) = \lim_{\Delta t \to 0} \frac{Prob\ (t\ \leq T < t + \Delta t, D = k\ |T \geq t\ \cup (T\ , t\ \cap K\ \neq k))}{\Delta t}$$

The subjects at risk for time t include those who have not experienced an event of type k and who may have experienced a competing event before time t.

2. Cause specific analysis

$$\lambda_k^{cs}(t) = \lim_{\Delta t \to 0} \frac{Prob\ (t\ \leq T < t + \Delta t, D = k\ |T \geq t\ )}{\Delta t}$$

The subjects at risk for time t therefore include those who have not experienced any types of events.

As shown above, subdistribution includes the subjects with the competing events into the risk set when calculating the probability of event at a certain time, this thus did not give a correct causal inference on the covariates. Therefore, the cause-specific hazard function is recommended when the purpose is on causal inference of covariates effect.

DATA EXAMPLE

In this data example, we compared two different arch replacement surgical strategies – aggressive arch replacement versus conservative hemiarch replacement for patients with acute type A aortic dissection. Event of interest is reoperation risk, competing risk is death, time origin is surgery date, censoring events include loss of follow up and end of study period. In the later model, we also adjusted for risk factors including age, gender, connective tissue disease status, sever AI condition, and hypertension.

Table 1. Data example variables

Variable name	Variable Meaning
time_reop_arch	Time variable denotes the event time or censor time since surgery
status	0 indicates censor without any event; 1 indicates reoperation; 2 indicates death before arch reoperation
group	1='Aggressive arch replacement' 0='Hemiarch replacement'
sever_AI	Severe aorta insufficiency
age_at_operation	Age at the time of initial operation
gender	Gender 1=female, 0=male
mfs_connect_tissue	Connective tissue disease
htn	Hypertension

```
proc format;
      value group  1='Aggressive arch replacement' 0='Hemiarch replacement' ;
   run;
```

if we ignore the competing risk and perform Kaplan-Meier estimates in the presence of competing risk, we will be facing two issues below:

1. Using the Kaplan-Meier estimate to estimate incidence function creates upward biases in the estimation of incidence function.

The Kaplan-Meier estimate of incidence of reop within year 5 is (0.071 for group 0) and (0.105 for group 1), compared to the cumulative incidence from Fine and Grey's method (0.063 for group 0 and 0.093 for group 1 (Table 2).

2. The sum of Kaplan-Meier estimates of the incidence of each outcome exceed the Kaplan-Meier estimate of the incidence of composite outcome that combines all event types.

For example, in Table 2, the sum of Kaplan-Meier estimates of incidence of death and reoperation within year 5 (0.264 for group 0; 0.343 for group 1). This is greater than the Kaplan-Meier estimate of incidence of the composite outcome (0.250 for group 0 and 0.318 for group 1). The sum of CIF from Grey method is equal to the sum of CIF derived from KM.

Table 2. Comparisons of CIF from Kaplan-Meier method versus Grey method

Group	Types of events	CIF from KM	CIF from Grey method
0	ALL	0.250	0.250
1	ALL	0.318	0.318
0	death	0.193	0.1868
1	death	0.238	0.2251
0	reop	0.071	0.063
1	reop	0.105	0.093
0	Sum (death, reop)	0.264	0.250
1	Sum (death, reop)	0.343	0.318

CRUDE INCIDENCE OF REOPERATION IN THE PRESENCE OF COMPETING RISK

SAS has two equivalent ways to describe subdistribution curves: %CIF macro [4] and event codes function in PROC LIFETEST [5, 6].

Generate a CIF curve using SAS macro %CIF

```
%CIF (data=arch, time=time_reop_arch,status=status, event=1, censored=0,
group=group, options=plotcl,
 title= CIF macro Subdistribution method);
quit;
```



Generate a CIF curve using SAS LIFETEST procedure

```
   * subdistribution method using eventcode option;
proc lifetest data=arch plots=cif (test cl ) atrisk maxtime=18;
title 'Subdistribution method for reoperation risk';
time time_reop_arch*status(0) /eventcode=1;strata group;
run;
```

Summary of Failure Outcomes					
Stratum	group	Failed Events	Competing Events	Censored	Total
1	Aggressive arch replacement	15	41	94	150
2	Hemiarch replacement	34	88	200	322
Total		49	129	294	472

From result output, we could obtain the cumulative incidence rate over time. Here is an example output.

Stratum 1: group = 0			
time_reop_arch	Cumulative Incidence	Standard Error	95% Confidence Interval

0	0	0	.	.
0.364384	0.00318	0.00318	0.000308	0.0167
0.419178	0.00636	0.00449	0.00129	0.0213
0.452055	0.00954	0.00549	0.00266	0.026
0.465753	0.0127	0.00633	0.00427	0.0305
0.471233	0.0159	0.00707	0.00604	0.0349
0.531507	0.0191	0.00774	0.00793	0.0392
……				



HAZARD MODELS

SAS provided two different model strategies for competing risk analysis. Subditribution method and cause-specific method. Previous study suggested that subditribution hazard models are more suitable for developing clinical prediction models, while cause-specific hazard models are more suitable for analyzing etiology questions.

Subdistribution hazard model

To fit a subdistribution model, we could use eventcode option in the model statement in PHREG procedure. Here, event code=1 indicated that reoperation is the event of interest, 0 is alive without reoperation, and coding 2 is the competing risk of death. For this Fine and Gray model, you could predict CIFs for the event using BASLINE statement, but model checking is not available in PHREG. One may use the log minus log of the subdistribution hazard or the Schoenfeld residuals for model checking [3].

```
* subdistribution using PHREG;
proc phreg data=arch plots(overlay=bystratum)=cif ;
class group (ref="0") gender  sever_AI(ref="0") mfs_connect_tissue (ref="0")
htn (ref="0");
model time_reop_arch*status(0)=group age_at_operation gender  sever_AI
mfs_connect_tissue htn/eventcode=1;
hazardratio group/diff=ref;
hazardratio age_at_operation/units=10;
hazardratio gender/diff=ref;
hazardratio sever_AI/diff=ref;
hazardratio mfs_connect_tissue/diff=ref;
hazardratio htn/diff=ref;
run;
```

Cause specific hazard model

To fit a cause specific hazard model, the competing risk is treated as a censoring event, so status (0,2) indicated that both alive without reoperation, and death before any reoperation are treated as censoring in the model. Treating all competing events as censoring ensures that the risk set at each event time contains only those subjects who did not experience any competing events or are truly censored. The existing tools such as ASSESS statement can be used to check the cause-specific Cox models. Starting in SAS/STAT 14.3, you may also use EVENTCODE (COX)=option in the MODEL statement to fit the cause-specific Cox models.

```
* cause-specific using PHREG;
proc phreg data=arch;
class group (ref="0") gender  sever_AI(ref="0") mfs_connect_tissue (ref="0")
htn (ref="0");
model time_reop_arch*status(0,2)=group age_at_operation gender  sever_AI
mfs_connect_tissue htn;
hazardratio group/diff=ref;
hazardratio age_at_operation/units=10;
hazardratio gender/diff=ref;
hazardratio sever_AI/diff=ref;
hazardratio mfs_connect_tissue/diff=ref;
hazardratio htn/diff=ref;
run;
```

COMPARISONS OF THE TWO METHODS

The table below provides the results that compare subdistribution model versus cause-specific hazard model for reoperation given death as a competing risk factor. We also compare to a regular COX model for death, as reoperation is not a competing risk for death.

Methods	Subdistribution	Cause-Specific	Regular COX
Event of interest	Reoperation	Reoperation	Death

Risk factors	Hazard ratio	95% Wald Confidence Limits		Point Estimate	95% Wald Confidence Limits		Hazard ratio	95% Wald Confidence Limits	
Group 1 vs 0	0.88	0.47	1.62	0.87	0.47	1.62	1.16	0.80	1.68
Age at operation Unit=1	0.97	0.96	1.00	0.98	0.96	1.01	1.04	1.03	1.06
Age at operation Unit=10	0.77	0.63	0.95	0.84	0.66	1.06	1.51	1.29	1.76
Gender 1 vs 2	1.46	0.72	2.99	1.49	0.73	3.04	1.06	0.72	1.56
Sever_AI 1 vs 0	0.46	0.23	0.91	0.43	0.19	0.95	1.05	0.70	1.59
Connect tissue disease 1 vs 0	1.13	0.51	2.49	1.22	0.43	3.48	1.52	0.63	3.67
Hypertension 1 vs 0	1.10	0.60	2.02	1.09	0.59	2.03	1.03	0.69	1.54

The quantity of hazard ratios from the two different methods are very similar in our example. But the interpretation is distinct. For example, aggressive arch replacement compared to hemiarch replacement decreased the relative incidence of reoperation by 12% (HR=0.88, 95%CI (0.47, 1.62), whereas aggressive arch decreased the cause-specific hazard of reoperation by 13%(HR=0.87, 95%CI (0.47, 1.62), For another example, a 10-year increase in age decreased the relative incidence of reoperation by 23% (HR=0.77, 95% CI (0.63, 0.95)), while it decreased cause-specific hazard of reoperation by 16% (HR=0.84, 95%CI (0.66, 1.06)). In contrast, age is a more pronounced risk factor for death. A 10-year increase in age increases the hazard of death by 51% (HR=1.51, 95% CI (1.29, 1.76)).

Noticeably a strong prognostic factor such as age for the hazard for the competing risk death has led to an apparent decrease in the cumulative incidence for reoperation when such factor has smaller effect on the cause specific hazard for reoperation.

CONCLUSION

This paper demonstrates the use of cumulative incidence function and cause-specific hazard function in time-to-event analysis adjusting for competing risk events.



REFERENCES

1. Austin, P.C., D.S. Lee, and J.P. Fine, *Introduction to the Analysis of Survival Data in the Presence of Competing Risks.* Circulation, 2016. **133**(6): p. 601-9.
2. John P. Klein , M.L.M., *Survival analysis techniques for censored and truncated data*. 2003.
3. Fine, J.P., and Gray, R. J., *A Proportional Hazards Model for the Subdistribution of a Competing Risk.* Journal of the American Statistical Association, 1999. **94**: p. 496–509.
4. Guixian Lin, Y.S., Gordon Johnston, *Analyzing Survival Data with Competing Risks Using SAS® Software.* SAS Global Forum 2012, 2012.
5. Ying So, G.L., and Gordon Johnston, *Using the PHREG Procedure to Analyze Competing-Risks Data.* SAS Institute Inc., 2014.
6. So, C.G.a.Y., *Cause-Specific Analysis of Competing Risks Using the PHREG Procedure.* SAS Institute Inc., 2018.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Xiaoting Wu (Ting), PhD, MS

Department of Cardiac Surgery

1500 E Medical Center Drive

Ann Arbor, MI 48109

734.936.7731

xiaotinw@med.umich.edu

Other brand and product names are trademarks of their respective companies.